/**
 * Copyright (c) 2008 - TheNittyGritty - Thomas Brekelmans
 * 
 * Permission is hereby granted, free of charge, to any person obtaining a copy
 * of this software and associated documentation files (the "Software"), to deal
 * in the Software without restriction, including without limitation the rights
 * to use, copy, modify, merge, publish, distribute, sublicense, and/or sell
 * copies of the Software, and to permit persons to whom the Software is
 * furnished to do so, subject to the following conditions:
 * 
 * The above copyright notice and this permission notice shall be included in
 * all copies or substantial portions of the Software.
 * 
 * THE SOFTWARE IS PROVIDED "AS IS", WITHOUT WARRANTY OF ANY KIND, EXPRESS OR
 * IMPLIED, INCLUDING BUT NOT LIMITED TO THE WARRANTIES OF MERCHANTABILITY,
 * FITNESS FOR A PARTICULAR PURPOSE AND NONINFRINGEMENT. IN NO EVENT SHALL THE
 * AUTHORS OR COPYRIGHT HOLDERS BE LIABLE FOR ANY CLAIM, DAMAGES OR OTHER
 * LIABILITY, WHETHER IN AN ACTION OF CONTRACT, TORT OR OTHERWISE, ARISING FROM,
 * OUT OF OR IN CONNECTION WITH THE SOFTWARE OR THE USE OR OTHER DEALINGS IN
 * THE SOFTWARE.
 */
package nl.thenittygritty.utils 
{
	
	/**
	 * ObjectUtils is a library class which shouldn't be instantiated. Instead 
	 * it provides static methods which help in working with Objects.
	 * 
	 * <p>ArrayUtils and ObjectUtils share the same duplicate method.</p>
	 */
	public class ObjectUtils 
	{
		/**
		 * Private constructor for the ObjectUtils, this shouldn't be called.
		 */
		public function ObjectUtils() 
		{
			// Throw an error here, we shouldn't try to instantiate ObjectUtils.
			throw new Error("The ObjectUtils shouldn't be instantiated.");
		}
		
		/**
		 * Compares two objects to each other.
		 */
		public static function compareObjects(objectA:Object, 
											  objectB:Object):Boolean 
		{
			for(var sItem:String in objectA) 
			{
				if(objectA[sItem] is Object) 
				{
					if(!ObjectUtils.compareObjects(objectA[sItem], 
												   objectB[sItem])) 
					{
						return false;
					}
				}
        		else 
				{
					if(objectA[sItem] != objectB[sItem]) 
					{
						return false;
					}
				}
			}
			return true;
		}
		
		/**
		 * Duplicates an object.
		 */
		public static function duplicate(source:Object, 
									duplicateRecursively:Boolean = false):Object 
		{
			var copy:Object;
			if(duplicateRecursively) 
			{
				if(source is Array) 
				{
					copy = new Array();
					for(var i:Number = 0; i < source.length; i++) 
					{
						if(source[i] is Object) 
						{
							copy[i] = duplicate(source[i]);
						}
            			else 
						{
							copy[i] = source[i];
						}
					}
					return copy;
				}
        		else 
				{
					copy = new Object();
					for(var item:String in source) 
					{
						if(source[item] is Object && 
						   !(source[item] is String) && 
						   !(source[item] is Boolean) && 
						   !(source[item] is Number)) 
						{
							copy[item] = duplicate(source[item], 
												   duplicateRecursively);
						}
            			else 
						{
							copy[item] = source[item];
						}
					}
					return copy;
				}
			}
      		else 
			{
				if(source is Array) 
				{
					return (source as Array).concat();
				}
        		else 
				{
					copy = new Object();
					for(var s:String in source) 
					{
						copy[s] = source[s];
					}
					return copy;
				}
			}
		}
	}
}
